RESEARCH NOTE NC-79

NORTH CENTRAL FOREST EXPERIMENT STATION, FOREST SERVICE—U.S. DEPARTMENT OF AGRICULTURE
Folwell Avenue, St. Paul, Minnesota 55101

Computer Calculation of Fire Danger

ABSTRACT. — This paper describes a computer program that calculates National Fire Danger Rating Indexes. Fuel moisture, buildup index, and drying factor are also available. The program is written in FORTRAN and is usable on even the smallest compiler.

OXFORD: 431.5:U681.3

Since its introduction in 1964 the National Fire Danger Rating System has been adopted by most Federal and State forest fire control agencies, and has become an integral part of the decision-making process in fire control planning. Modern high-speed computers are an efficient tool for manipulating and analyzing the vast amount of fire danger data generated from this system. Using the National Fire Danger Rating System requires working with as many as five tables and seven variables. Danger ratings are obtained from this tabular method on a day-to-day basis. However, when mass conversion of historical weather data to the present system is required, the computer is an essential tool. In such cases a computer solution will speed up the calculations and provide computational accuracy.

This paper describes a computer program that computes buildup index, fire-load index, and various spread indexes. The program closely follows the method detailed by Nelson.[1] An equation computation method was chosen instead of a table look-up scheme for this program.[2] Primarily, the decision was dictated by computer memory limitations. Utilizing a look-up method would necessitate larger core size because of the 1,218 storage locations needed for the four tables.

The results from the formulas differ slightly from the table values. However, the differences are not great, and there is reason to believe that the computed results are more accurate than the tabular results.[3]

The flow chart (fig. 1) and listing of the program are included so the program may be used on the smallest FORTRAN compiler. The program has been thoroughly tested and run on an IBM 360/30 and requires less than 4K of memory. It uses the lowest possible level of FORTRAN, thus ensuring success on any machine.

For input, the program requires dry-bulb and wet-bulb readings, a yes or no decision regarding snow on the ground, the preceding 24-hour precipitation, the current windspeed, yesterday's buildup index, and current herbaceous stage of vegetation. These are the same variables required by the tabular method.

---

[1] *Nelson, Ralph M. The National Fire Danger Rating System: derivation of spread index for eastern and southern States. U.S.D.A. Forest Serv. Res. Pap. SE-13, 44 p., illus. (Southeast. Forest Exp. Sta., Asheville, N.C.). 1964.*

[2] *Barney, Richard J. Calculating the National Fire-Danger Rating Spread Index by computer. U.S.D.A. Forest Serv. Res. Note INT-19, 2 p. (Intermountain Forest, Range Exp. Sta., Ogden, Utah.) 1964.*

[3] *Memorandum from John J. Keetch dated December 28, 1962. On file at North Central Forest Exp. Sta., St. Paul, Minn.*

The program first tests for snow on the ground, a condition that results in spread index values of zero. If there is no snow, it calculates the fine fuel moisture and drying factor. Yesterday's buildup index is then adjusted for precipitation and the drying factor is added. Timber fuel moisture content is adjusted from the current buildup index. The fine fuel spread index and timber spread index are then calculated. The last step computes the fire-load rating, man-hour base.

The program calculates fine fuel spread, timber spread, buildup index recovery, and fire-load rating for each day. Additionally, fine fuel moisture, adjusted fuel moisture, and drying factor are also available, if desired.

The following formulas have been circulated in memoranda but have never been published. The formulas appear in the program and are reproduced here for greater emphasis and clarity.

Fine Fuel Moisture[4]

$$FFM = Ae^{B} \text{ (dry-wet)}$$

Where

A	B	Range of (dry-wet)
30.0	−.1859	< 4.5° F.
19.2	−.0859	< 12.5° F.
13.8	−.0579	< 27.5° F.
22.5	−.0774	> 27.5° F.

A and B are the piecewise regression coefficients used to determine FFM. The depression of the wet bulb is used to decide which set of A and B will be used. Herb stage is used to adjust the calculated fine fuel moisture by adding 5 percent for transition stage or 10 percent for green fuels.

Adjusted Fuel Moisture[5]

(50-Day Lag)

$$ADFM = .9FFM + 9.5e^{\frac{-BUI}{50.}}$$

Where — FFM is the current fine fuel moisture

BUI is today's buildup index recovery

e is the base of the natural logs

Adjusted fuel moisture is also equal to equivalent fuel moisture.

Buildup Index[3]

$$BUI = -50\ (\log_e (1-(-e^{\frac{BUO}{50.}}))_e 1.175\ (PRECIP - .1))$$

Where — BUI is today's buildup recovery corrected for any precipitation in the past 24 hours

BUO is yesterday's buildup

PRECIP is the past 24 hours precipitation in inches and hundredths

LOG_e is the natural log

is the base of natural logs

BUI must be adjusted by adding the drying factor *after* correction for any precipitation greater than 0.1 inch.

Fine Fuel Spread[5]

$$GRASS = A\ (WIND + B)\ (33. - FFM)^{1.65} - 3.$$

Timber Spread Index[5]

$$TIMBER = A\ (WIND + B)\ (33. - ADFM)^{1.65} - 3.$$

Where	A	B	Windspeed
	0.01312	6.0	<14 m.p.h.
	0.009184	14.4	>14 m.p.h.

FFM = Fine fuel moisture

ADFM = Fuel moisture adjusted for 50-day timelag

WIND = Windspeed in m.p.h., 20 feet above open level ground

Fire Load Index[6]

(Man-Hour Base)

$$FLOAD = 10.^{(1.75(\text{Log}_{10}\,TIMBER) + .32\,(\text{Log}_{10}\,BUI) - 1.64)}$$

Where — TIMBER is the Timber Spread index

BUI is the buildup index recovery

Fire load index is not a feature of the National system, but has been suggested as an experimental feature for consideration.

---

[4] *Memorandum from George M. Byram dated January 30, 1963. On file at North Central Forest Exp. Sta., St. Paul, Minn.*

[5] *Memorandum from David Bruce dated March 1, 1963. On file at North Central Forest Exp. Sta., St. Paul, Minn.*

[6] *Outline for discussion of fire load rating, prepared by John J. Keetch, undated. On file at North Central Forest Exp. Sta., St. Paul, Minn.*



Figure 1. — Flow chart of subroutine to calculate National Fire Danger Rating Indexes.

Program

```
      SUBROUTINE DANGER (DRY,WET,ISNOW,PRECIP,WIND,BUO,IHERB,          DNGR 001
     /      DF,FFM,ADFM,GRASS,TIMBER,FLOAD)                           DNGR 002
C     ROUTINE FOR COMPUTING NATIONAL FIRE DANGER RATINGS AND FIRE LOAD INDEX
C     DATA NEEDED FOR THE CALCULATIONS ARE=
C     DRY,     DRY BULB TEMPERATURE
C     WET,     WET BULB TEMPERATURE
C     ISNOW,   SOME POSITIVE NON ZERO NUMBER IF THERE IS SNOW ON THE GROUND
C     WIND,    THE CURRENT WIND SPEED IN MILES PER HOUR
C     BUO,     THE LAST VALUE OF THE BUILD UP INDEX
C     IHERB,   THE CURRENT HERB STATE OF THE DISTRICT 1=CURED,2=TRANSITION,3=GREEN
C     DATA RETURNED FROM THE SUBROUTINE ARE
C     DRYING FACTOR AS                          DF
C     FINE FUEL MOISTURE AS                     FFM
C     ADJUSTED (10 DAY LAG) FUEL MOISTURE AS    ADFM
C     GRASS SPREAD INDEX WILL BE RETURNED AS    GRASS
C     TIMBER SPREAD INDEX WILL BE RETURNED AS   TIMBER
C     FIRE LOAD RATING (MAN-HOUR BASE) AS       FLOAD
C     BUILD UP INDEX WILL BE RETURNED AS        BUO
      DIMENSION  A(4),B(4),C(3),D(6)                                   DNGR 003
      FFM= 99.                                                         DNGR 004
      ADFM= 99.                                                        DNGR 005
      DF=0.                                                            DNGR 006
      FLOAD=0.                                                         DNGR 007
C     THESE ARE THE TABLE VALUES USED IN COMPUTING THE DANGER RATINGS
      A(1) = -0.185900                                                 DNGR 008
      A(2) = -0.85900                                                  DNGR 009
      A(3) = -0.059660                                                 DNGR 010
      A(4) = -0.077373                                                 DNGR 011
      B(1)=30.0                                                        DNGR 012
      B(2)=19.2                                                        DNGR 013
      B(3)=13.8                                                        DNGR 014
      B(4)=22.5                                                        DNGR 015
      C(1) = 4.5                                                       DNGR 016
      C(2) = 12.5                                                      DNGR 017
      C(3) = 27.5                                                      DNGR 018
      D(1)=16.0                                                        DNGR 019
      D(2)=10.0                                                        DNGR 020
      D(3)= 7.0                                                        DNGR 021
      D(4)= 5.0                                                        DNGR 022
      D(5)= 4.0                                                        DNGR 023
      D(6)= 3.0                                                        DNGR 024
C     TEST TO SEE IF THERE IS SNOW ON THE GROUND

      IF(ISNOW) 5,5,1                                                  DNGR 025
C     THERE IS SNOW ON THE GROUND AND THE TIMBER AND GRASS SPREAD INDEXES
C     MUST BE SET TO ZERO.  WITH A ZERO TIMBER SPREAD THE FIRE LOAD IS
C     ALSO ZERO.  BUILD UP WILL BE ADJUSTED FOR PRECIPITATION.
    1 GRASS=0.                                                         DNGR 026
      TIMBER=0.                                                        DNGR 027
      IF ( PRECIP - .1 ) 4,4,2                                         DNGR 028
C     PRECIPITATION EXCEEDED  .1 INCHES AND WE REDUCE THE BUILD UP INDEX
    2 BUO=-50.*ALOG(1.-(1.-EXP (-BUO/50.))*EXP (  -1.175*(PRECIP-.1)))  DNGR 029
      IF ( BUO ) 3,4,4                                                 DNGR 030
    3 BUO=0.                                                           DNGR 031
    4 RETURN                                                           DNGR 032
C     THERE IS NO SNOW ON THE GROUND AND WE WILL COMPUTE THE SPREAD INDEXES
C     AND FIRE LOAD
    5 DIF=DRY-WET                                                      DNGR 033
      DO 6 I=1,3                                                       DNGR 034
      IF( DIF-C(I) ) 7,7,6                                             DNGR 035
    6 CONTINUE                                                         DNGR 036
      I=4                                                              DNGR 037
    7 FFM=B(I)*EXP (A(I)*DIF)                                          DNGR 038
C     WE WILL  NOW FIND THE DRYING FACTOR FOR THE DAY
      DO 8 I=1,6                                                       DNGR 039
      IF ( FFM - D(I) ) 8,8,9                                          DNGR 040
    8 CONTINUE                                                         DNGR 041
      DF=7                                                             DNGR 042
      GO TO 10                                                         DNGR 043
    9 DF=I-1                                                           DNGR 044
C     TEST TO SEE IF THE FINE FUEL MOISTURE IS ONE OR LESS
C     IF FINE FUEL MOISTURE IS ONE OR LESS WE SET IT TO ONE
   10 IF ( FFM-1. ) 11,12,12                                           DNGR 045
   11 FFM=1.                                                           DNGR 046
C     ADD 5 PERCENT FINE FUEL MOISTURE FOR EACH HERB STAGE GREATER THAN ONE
   12 FFM = FFM + ( IHERB-1 ) * 5.                                     DNGR 047
C     WE MUST ADJUST THE BUI FOR PRECIPITATION BEFORE ADDING THE DRYING FACTOR
      IF (PRECIP -.1) 15,15,13                                         DNGR 048

C     PRECIPITATION EXCEEDED  0.10 INCHES      WE MUST REDUCE THE
C     BUILD UP INDEX (BUO) BY AN AMOUNT EQUAL TO THE RAIN FALL
   13 BUO=-50.*ALOG(1.-(1.-EXP (-BUO/50.))*EXP (-1.175*(PRECIP-.1)))   DNGR 049
      IF ( BUO ) 14,15,15                                              DNGR 050
   14 BUO=0.0                                                          DNGR 051
C     AFTER CORRECTION FOR RAIN, IF ANY, WE ARE READY TO ADD TODAY'S
C     DRYING FACTOR TO OBTAIN THE CURRENT BUILD UP INDEX
   15 BUO=BUO+DF                                                       DNGR 052
C     WE WILL ADJUST THE GRASS SPREAD INDEX FOR HEAVY FUEL LAGS
C     THE RESULT WILL BE THE TIMBER SPREAD INDEX
C     THE ADJUSTED FUEL MOISTURE, ADFM, ADJUSTED FOR HEAVY FUELS, WILL
C     NOW BE COMPUTED
      ADFM = .9*FFM +.5 +9.5*EXP ( -BUO/50.)                           DNGR 053
C     TEST TO SEE IF THE FUEL MOISTURES ARE GREATER THAN 30 PERCENT.
C     IF THEY ARE, SET THEIR INDEX VALUES TO 1.
      IF ( ADFM-30. ) 19,16,16                                         DNGR 054
   16 IF (  FFM-30. ) 18,17,17                                         DNGR 055
C     FINE FUEL MOISTURE IS GREATER THAN 30 PERCENT, THUS WE SET THE GRASS
C     AND TIMBER SPREAD INDEXES TO ONE.
   17 GRASS = 1.                                                       DNGR 056
      TIMBER = 1.                                                      DNGR 057
      RETURN                                                           DNGR 058
   18 TIMBER = 1.                                                      DNGR 059
C     TEST TO SEE IF THE WIND SPEED IS GREATER THAN 14 MPH
      IF ( WIND-14. ) 21,25,25                                         DNGR 060
   19 IF ( WIND-14. ) 20,24,24                                         DNGR 061
   20 TIMBER = .01312*(WIND+6.) * (33.-ADFM)**1.65 - 3.                DNGR 062
   21 GRASS  = .01312*(WIND+6.) * (33.- FFM)**1.65 - 3.                DNGR 063
      IF ( TIMBER-1. ) 22,22,28                                        DNGR 064
   22 TIMBER = 1.                                                      DNGR 065
      IF ( GRASS-1. ) 23,28,28                                         DNGR 066
   23 GRASS = 1.                                                       DNGR 067
      GO TO 28                                                         DNGR 068
C     WIND SPEED IS GREATER THAN 14 MPH.  WE USE A DIFFERENT FORMULA.
   24 TIMBER = .00918*(WIND+14.) * (33.-ADFM)**1.65 - 3.               DNGR 069
   25 GRASS  = .00918*(WIND+14.) * (33.- FFM)**1.65 - 3.               DNGR 070
      IF ( GRASS-99. ) 28,28,26                                        DNGR 071


   26 GRASS = 99.                                                      DNGR 072
      IF ( TIMBER-99. ) 28,28,27                                       DNGR 073
   27 TIMBER = 99.                                                     DNGR 074
C     WE HAVE NOW COMPUTED THE GRASS AND TIMBER SPREAD INDEXES
C     OF THE NATIONAL FIRE DANGER RATING SYSTEM.  WE HAVE THE
C     BUILD UP INDEX AND NOW WE WILL COMPUTE THE FIRE LOAD RATING
   28 IF ( TIMBER ) 30,30,29                                           DNGR 075
   29 IF ( BUO ) 30,30,31                                              DNGR 076
C     IT IS NECESSARY THAT NEITHER TIMBER SPREAD NOR BUILD UP BE ZERO
C     IF EITHER TIMBER SPREAD OR BUILD UP  IS ZERO, FIRE LOAD IS ZERO
   30 RETURN                                                           DNGR 077
C     BOTH TIMBER SPREAD AND BUILD UP ARE GREATER THAN ZERO
   31 FLOAD=1.75*ALOG10( TIMBER ) + .32*ALOG10( BUO ) - 1.640          DNGR 078
C     ENSURE THAT FLOAD IS GREATER THAN ZERO, OTHERWISE SET IT TO ZERO.
      IF ( FLOAD ) 32,32,33                                            DNGR 079
   32 FLOAD = 0.                                                       DNGR 080
      RETURN                                                           DNGR 081
   33 FLOAD = 10. ** FLOAD                                             DNGR 082
      RETURN                                                           DNGR 083
      END                                                              DNGR 084
```

WILLIAM A. MAIN
Forest Research Technician

1969